Slika:Pm1234 Ground.png

Iz Wikipedije, proste enciklopedije

Logo Wikimedijine zbirke Datoteka je del Wikimedijine zbirke. Najverjetneje so vsi navedeni podatki s tamkajšnje opisne strani.

[edit] Summary

Terms and partial sums of 1 − 2 + 3 − 4 + · · · going out to the horizon.

  • The terms are indicated as black lines executing displacements to the camera's right, plus a constant displacement away from the camera.
  • The partial sums are where the terms end; they are indicated as black circles.
  • The action takes place on an infinite horizontal plane. The integers are located on the gray lines. The horizon is picked out by a bluish sky.
  • The camera type is rectangular perspective. Of course, it is placed over x = 1/4 for symmetry.

[edit] POV-Ray source

Warning: this scene requires a lot of antialiasing, and it is slow.

camera { 
        location <.25, 4, -4.5>
        up y*image_height right x*image_width angle 65 look_at <.25, 0, 10>
}
background { rgb 1 }

#declare an = 0;
#declare sn = 0;
#declare snn = 0;
union {
        #declare n=1;
        #while (n<15000)
                #declare an = n * pow(-1, n+1);
                #declare snn = sn;
                #declare sn = sn + an;
                cylinder { 0, z*sqrt(1+an*an), .1 scale <1, 0.3/sqrt(n), 1>
                        rotate y*degrees(atan(an)) translate <snn, 0, n>
                }
                sphere { <snn, 0, n>, .2 scale <1, 0.3/sqrt(n), 1> }
                #declare n=n+1;
        #end
        pigment { color rgb 0 }
}

intersection {
        plane { y, .01/(n*n) pigment { color rgb 0 } }
        plane { -z, -n  }
        plane { -x, 0 rotate y*-degrees (atan (.5)) translate <0, 0, 1> }
        plane { x, 0 rotate y*degrees (atan (.5)) translate <1, 0, 2> }
        pigment { color rgbt 1 }
}

plane {
        y, 0
        pigment {
                gradient x
                color_map { 
                        [0 rgbt .5]
                        [0.024 rgbt .5]
                        [0.025 rgbt 1]
                        [0.975 rgbt 1]
                        [0.976 rgbt .5]
                        [1 rgbt .5]
      }
        }
}
plane { -y, -5 pigment { color rgbf <.9, .9, 1, .9> } }

[edit] Licensing

I, the author of this work, hereby publish it under the following licenses:
GNU head Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.2 or any later version published by the Free Software Foundation; with no Invariant Sections, no Front-Cover Texts, and no Back-Cover Texts. A copy of the license is included in the section entitled "GNU Free Documentation License".

العربية | Asturianu | Български | বাংলা | ইমার ঠার | Brezhoneg | Bosanski | Català | Sinugboanong Binisaya | Česky | Dansk | Deutsch | Ελληνικά | English | Esperanto | Español | Eesti | Euskara | فارسی | Suomi | Français | Galego | עברית | Hrvatski | Magyar | Bahasa Indonesia | Ido | Íslenska | Italiano | 日本語 | ქართული | 한국어 | Kurdî / كوردي | Latina | Lëtzebuergesch | Lietuvių | Bahasa Melayu | Nnapulitano | Nederlands | Norsk (nynorsk) | Norsk (bokmål) | Polski | Português | Română | Русский | Slovenčina | Slovenščina | Shqip | Српски | Svenska | తెలుగు | ไทย | Türkçe | Українська | اردو | Tiếng Việt | Volapük | 中文(简体) | 中文(繁體) | +/-

Some rights reserved
Creative Commons Attribution iconCreative Commons Share Alike icon
This file is licensed under the Creative Commons Attribution ShareAlike license versions 2.5, 2.0, and 1.0

العربية | Česky | Dansk | Deutsch | English | Español | فارسی | Français | Italiano | 日本語 | Nederlands | Polski | Português | Русский | Svenska | Türkçe | 简体中文 | 正體中文 | +/-

You may select the license of your choice.

Datoteka je del naslednjih strani slovenske Wikipedije (strani drugih projektov niso navedene):